



Unidade de Educação Profissional do Recife

Estrutura de Repetição RECIFE, 2024



Laços de Repetição

A estrutura de repetição é um bloco de códigos que se repete diversas vezes até que uma condição se torne falsa.

Existe 4 tipos de estruturas de repetição

para **(condição de parada)**

enquanto **(condição de parada)**

faça - enquanto **(condição de parada)**

enquanto **(verdadeiro)**

Como funciona a estrutura PARA?

A estrutura **para** tem três condições principais:

Sintaxe do para

para(início; parada; incremento)

início: é um valor que corresponde a partir de qual número o intervalo será iniciado;

parada: é um valor que indica o número de parada do intervalo;

incremento: é um valor indica o valor que queremos adicionar entre um item e outro.

Exemplo 1 - PARA

Escrever um pseudocódigo para exibir os números de 1 até 10 na tela.

```
programa
{
    funcao inicio()
    {
        para(inteiro contadora=0; contadora <= 10; contadora++){
            escreva(contadora, " ")
        }
    }
}
```

0 1 2 3 4 5 6 7 8 9 10

Exemplo 2 - PARA

Escrever um pseudocódigo que mostre uma multiplicação da tabuada do número 1 até a tabuada do número 10.

```
programa
{
    funcao inicio()
    {
        para(inteiro linha=1; linha <= 10; linha++){
            escreva("Tabuada ", linha, "\n")
            para(inteiro coluna = 0 ; coluna <=10; coluna++){
                escreva(linha, "X", coluna,"=", linha*coluna, "\n")
            }
        }
    }
}
```

```
1X0=0
1X1=1
1X2=2
1X3=3
1X4=4
1X5=5
1X6=6
1X7=7
1X8=8
1X9=9
1X10=10
```



Comando: enquanto

Como funciona a estrutura ENQUANTO?

O laço **enquanto** executa um ou mais comandos enquanto a condição seja verdadeira e precisa ter um contador de parada.

```
programa
{
    funcao inicio()
    {
        inteiro contador = 1
        enquanto(contador <= 4){
            escreva(contador, " ")
            contador++
        }
    }
}
```

Precisamos de uma **variável de apoio(contador)** para indicar o ponto de parada da estrutura de repetição enquanto.

1 2 3 4

Caso o programa não tenha incremento ficará em **“loop infinito”**

Exemplo 1: enquanto

Escrever um pseudocódigo que peça o nome de duas pessoas e mostrar na tela.

```
funcao inicio()
{
    inteiro contador = 1
    cadeia nome = " "
    enquanto(contador <= 2){
        escreva("Informe um nome: ")
        leia(nome)
        contador++
    }
    escreva("O nome é: ", nome)
}
```

Neste caso, o nome só foi impresso o último porque a variável **nome** é uma variável simples, ou seja, só recebe um valor por vez.

```
Informe um nome: ana
Informe um nome: maria
O nome é: maria
```


Comando: faça enquanto

Como funciona a estrutura FACA ENQUANTO?

O laço **faca enquanto** executa um bloco de código enquanto a condição seja verdadeira.

```
funcao inicio()  
{  
    cadeia nome = ""  
    inteiro contador=0  
    faca{  
        escreva("Informe seu nome: ")  
        leia(nome)  
        escreva(nome)  
        contador++  
    }enquanto(contador<2)  
}
```

```
Informe um nome: ana  
Informe um nome: maria  
0 nome é: maria
```

Comando:
enquanto(verdadeiro)

Comando auxiliar para estrutura de repetição

O **enquanto(verdadeiro)** é um loop infinito proposital que não utiliza contador.

Para parar o loop infinito existe um comando chamado de PARE.

pare: É um comando de interrupção de um loop. Não importa o estado que se encontra a execução, ele parará o loop imediatamente.

Como funciona a estrutura ENQUANTO INFINITO?

Neste caso, o **enquanto** irá ser um loop infinito proposital sem necessitar de um contador.

```
funcao inicio()
{
    inteiro opcao = 0
    enquanto(verdadeiro){
        escreva("1- Para entrar no sistema \n")
        escreva("2- Para sair no sistema, digite: ")
        leia(opcao)
        se (opcao == 1){
            escreva("Bem vindo ao sistema \n")
        }
        senao{
            escreva("Até logo!")
            pare
        }
    }
}
```

```
1- Para entrar no sistema
2- Para sair no sistema, digite: 2
Até logo!
```

Arredondamentos de números reais

Arredondar números

No Portugol, arredondar um número real é muito fácil. Primeiro precisamos **incluir uma biblioteca chamada de Matematica** antes da função início.

Podemos renomear o nome da biblioteca para mat

programa

```
{
    inclui biblioteca Matematica --> mat

    funcao inicio()
    {

    }
}
```



Arredondar números

```
programa
```

```
{
```

```
  inclui biblioteca Matematica --> mat
```

```
  funcao inicio()
```

```
{
```

```
    real numero1, numero2, soma
```

```
    escreva("Informe o primeiro número: ")
```

```
    leia(numero1)
```

```
    escreva("Informe o segundo número: ")
```

```
    leia(numero2)
```

```
    soma = numero1 + numero2
```

```
    escreva("A soma arredondada é: ", mat.arredondar(soma, 3))
```

```
}
```

```
}
```

Bibliotecas sempre acima da função início.

Chama a biblioteca com o método de arredondar, em seguida, chama a variável e a quantidade de casas

Agora é sua vez!

Exercício
Estrutura de Repetição

Agora é sua vez!

1. Usando a estrutura **para** informe os números de 1 a 100.
2. Utilizando novamente a função **para** mostre os valores com início 0, condição de parada em 12, incrementando de 2 em 2.
3. Utilizando a estrutura **enquanto(contadora)** informe os números de 1 a 50.
4. Faça um programa que peça uma nota, entre zero e dez. Se o usuário digitar uma nota menor que zero ou maior que dez, deverá mostrar uma mensagem de erro, e pedir novamente a nota. Quando o usuário colocar uma nota válida, deverá mostrar o valor na tela e parar o loop. **(estrutura de repetição de sua preferência)**
5. Faça um programa que leia um nome de usuário e a sua senha. Não aceite a senha igual ao nome do usuário, caso isso ocorra, deve mostrar uma mensagem de erro e voltar a pedir as informações. **(estrutura de repetição de sua preferência)**